On the Verification of a Real Prolog Program: crossword

Roberto Bruni Marco Pistore

March 1996

1 The problem: the 4^{th} exercise of ILPS’94

An empty crossword puzzle is exemplified by the facts:

dimension(5). % it is always a square crossword puzzle
black(1,3). % meaning: square 1,3 is black
black(2,3). black(3,2). black(4,3). black(5,1). black(5,5).

These example facts represent the empty puzzle:

A list of words is also given, for example:

words([do,ore,ma,lis,ur,as,po,so,pirus,oker,al,adam,ik]).

All words in the list have at least two characters. The goal is to fill the puzzle with all the words exactly once, or to fail if that is impossible. For instance, filling the exemplified puzzle with the above words gives: A D A M

S O A L O I

P I R U S

O K E R The program should give the solution as the output parameter of crossword/1 and in the form of a list of words in the following order:

[as,po,do,ik,ore,ma,ur,lis,adam,so,al,pirus,oker]

i.e., first all the horizontal words, row by row and in a row as they occur from left to right, then all the vertical words, column by column.

If there is more than one solution, the program should produce them all by backtracking.

2 A solution

We present a solution to the problem, together with a specification (pre- and post-conditions and level mappings), a moding and a typing for each predicate.

We use the following types:

• GaeN are ground arithmetic expressions corresponding to natural numbers; if N ∈ GaeN then we use N itself to refer to val(N ).

• Char are characters.

• Atom are prolog atoms. For sake of simplicity, in what follows we identify an atom with the list of its characters.

• T erm are generic prolog terms.

• If T ype is a type, T ypeList are the finite lists of elements of type T ype. • Dir = {orz, vrt} are the directions in the crossword. • F ilaDim = {fila(D, Y, X, L) | D ∈ Dir ∧ Y, X, L ∈ GaeN } represents filling boxes of the crossword with direction D, starting cell (Y, X) and with length L. • F ila = {fila(D, Y, X, W s) | D ∈ Dir ∧ Y, X ∈ GaeN ∧ W s ∈ AtomList} represents filling boxes of the crossword with direction D, starting cell (Y, X); W s represents the set of the atoms which can be assigned to the particular box. • Res = {res(D, Y, X, W ) | D ∈ Dir ∧ Y, X ∈ GaeN ∧ W ∈ Atom} represents boxes of the crossword with direction D, starting cell (Y, X) and filled with atom W . • Cell = {cell(Y, X, Ch) | Y, X ∈ GaeN ∧ Ch ∈ Char}. • Cross = CellList. In the program we omit pre- and post-conditions when they are obvious. Moreover, we use some shorthands:

• If L is a list, {]L[} is the (multi)set of the elements of L.

• If SC is a set of Cell, SF is a set of F ila and SR is a set of Res, then SR ∈ sol(SC, SF ) if and only if either SF = ∅ = SR, or whenever R_0 = res(D, Y, X, W ) ∈ SR then: 1. F_0 = fila(D, Y, X, W s) ∈ SF and W ∈ {]W s[}, 2. SR−{R_0} ∈ sol(SC^0 , SF^0), where SC^0 = assign(SC, W, D, Y, X) and SF^0 = {fila(D, Y, X, W s^0) |

fila(D, Y, X, W s) ∈ SF − {F_0} ∧ {]W s^0[} = filter(D, Y, X, {]W s[} − {W }, SC^0)}.

• If SC is a set of Cell, W = [Ch_1, . . . , Ch_≪] ∈ Atom, D ∈ Dir and Y, X ∈ GaeN then assign(SC, W, D, Y, X) = æ_{SC ∪ {cell(Y, X + i, Ch⋌) | i = 0..n} if D = orz} SC ∪ {cell(Y + i, X, Ch_⋌) | i = 0..n} if D = vrt • If Y, X ∈ GaeN , SW is a set of Atom and SC is a set of Cell, then filter(orz, Y, X, SW, SC) = {[Ch_0, . . . , Ch_≪] ∈ SW | cell(Y, X+i, Ch) ∈ SC, i = 0..n =⇒ Ch = Ch_⋌ } and analogously for vrt. • If SC is a set of Cell then consistent(SC) holds iff cell(Y, X, Ch), cell(Y, X, Ch^0) ∈ SC imply Ch = Ch^0.

• If SF is a set of F ila and SC is a set of Cell then compatible(SF, SC) is true if and only if

fila(orz, Y, X, W s) ∈ SF , [Ch_0, . . .Ch_≪] ∈ {]W s[}, cell(Y, X + i, Ch) ∈ SC, i = 0..n imply

Ch = Ch_⋌; analogously for vrt.

• If R = res(D, Y, X, W ) and R^0 = res(D^0, Y^0, X^0, W^0) then R <_® R^0 iff one of the following conditions holds:

{ D = orz and D^0 = vrt; or { D = D^0 = orz and Y < Y^0; or { D = D^0 = orz, Y = Y^0 and X < X^0; or { D = D^0 = vrt and X < X^0; or { D = D^0 = vrt, X = X^0 and Y < Y^0.

• If C is the empty crossword correspoding to dimension n and set of black cells black, and Y, X = 1..n + 1 then orz(C, Y, X) gives the ordered list (left-to-right and top-down) of F ilaDim which describes the contiguous horizontal empty areas between black cells (and margins), from position (Y, X) towards the end of the crossword. For instance, if C is the empty puzzle previously described, then

orz(C, 3, 2) = [filadim(orz, 3, 2, 0), filadim(orz, 3, 3, 3), filadim(orz, 4, 1, 2),

filadim(orz, 4, 4, 2), filadim(orz, 5, 1, 0), filadim(orz, 5, 2, 3),
filadim(orz, 5, 6, 0), filadim(orz, 6, 1, 0)]

As this example shows, in the list do appear some areas on length 0 when two consecutive black cells (or a black cell and a margin) are present in the crossword and an area of length 0 for the (n + 1)^{th} row.

• vrt(C, Y, X) is defined similarly. • If F DS is a set of F ilaDim then useful(F DS) = {filadim(D, Y, X, L) | L > 1}. • A solution of the puzzle corresponding to crossword C and words ws is an ordered list W s of words that fill the crossword and such that {]W s[} ⊆ ws; it is an exact solution if {]W s[} = ws.

Suppose that crossword C_0 has dimension n_0 and black_0 as set of black cells; ws_0 is the set of words which should be used to fill C_0. We use P_0 to denote the puzzle corresponding to crossword C_0 and set of words ws_0.

%
% Crossword solver
%
% by Roberto BRUNI and Marco PISTORE
%
% (C) 1995
%

% dimension(− : GaeN )
% P re = {dimension(N )}
% P ost = {dimension(n_0)}
dimension(5).

% black(+ : GaeN, + : GaeN )
% P ost = {black(Y, X) | (Y, X) ∈ black_0}
black(1,3).
black(2,3).
black(3,2).

black(4,3).
black(5,1).
black(5,5).

% words(− : AtomList)
% P re = {words(W s)}
% P ost = {words(W s) | {]W s[} = ws_0}
words([do,ore,ma,lis,ur,as,po,so,pirus,oker,al,adam,ik]).

% crossword(− : AtomList)
% P ost = {crossword(W s) | W s is an exact solution of P_0 }
crossword(Ws) :-
    build_file(Fs),
    words(Ws1), ln(Ws1, L), ln(Fs, L),
    build_cross(Fs, [], Rs),
    ordina_res(Rs, Ws).

% build_file(− : F ilaList)
% P ost = {build_file(F s) | {R_1, . . . , R_≪} ∈ sol(∅, {]F s[}), R_⋌ <_® R_{⋌+1},
% R_⋌ = res(D_⋌ , Y_⋌ , X_⋌, W_⋌ ) =⇒ [W_1, . . . , W_≪] is a solution of P_0}
build_file(Fs) :-
    build_orz(1,1,FDs1),
    build_vrt(1,1,FDs2),
    app(FDs1,FDs2,FDs),
    assoc_words(FDs, Fs).

% build_orz(+ : GaeN, + : GaeN, − : F ilaDimList)
% P re = {build_orz(Y, X, F Ds) | Y, X ∈ [1..n_0 + 1]}
% P ost = {build_orz(Y, X, F Ds) | Y, X ∈ [1..n_0 + 1] ∧ F Ds = orz(C_0, Y, X)}
% |build_orz(Y, X, F Ds)| = (n_0 + 1 − X) + (n_0 + 1 − Y )(n_0 + 1)
build_orz(M,X,[filadim(orz,M,X,0)]) :-
    dimension(N), M =:= N+1.
build_orz(Y,M,[filadim(orz,Y,M,0) | FDs]) :-
    dimension(N), M =:= N+1, Y =< N, Y1 is Y + 1,
    build_orz(Y1,1,FDs).
build_orz(Y,X,[filadim(orz,Y,X,L1) | FDs]) :-
    dimension(N), X=<N, Y =<N,
    \+black(Y,X), X1 is X+1,
    build_orz(Y,X1,[filadim(_,_,_,L) | FDs]),
    L1 is L+1.
build_orz(Y,X,[filadim(orz,Y,X,0) | FDs]) :-
    dimension(N), X=<N, Y =<N,
    black(Y,X), X1 is X+1,
    build_orz(Y,X1,FDs).

% build_vrt(+ : GaeN, + : GaeN, − : F ilaDimList)
% P re = {build_vrt(Y, X, F Ds) | Y, X ∈ [1..n_0 + 1]}
% P ost = {build_vrt(Y, X, F Ds) | Y, X ∈ [1..n_0 + 1] ∧ F Ds = vrt(C_0, Y, X)}
% |build_vrt(Y, X, F Ds)| = (n_0 + 1 − Y ) + (n_0 + 1 − X)(n_0 + 1)
build_vrt(Y,M,[filadim(vrt,Y,M,0)]) :-
    dimension(N), M =:= N+1.
build_vrt(M,X,[filadim(vrt,M,X,0) | FDs]) :-
    dimension(N), M =:= N+1, X =< N, X1 is X + 1,

    build_vrt(1,X1,FDs).
build_vrt(Y,X,[filadim(vrt,Y,X,L1) | FDs]) :-
    dimension(N), X=<N, Y =<N,
    \+black(Y,X), Y1 is Y+1,
    build_vrt(Y1,X,[filadim(_,_,_,L) | FDs]),
    L1 is L+1.
build_vrt(Y,X,[filadim(vrt,Y,X,0) | FDs]) :-
    dimension(N), X=<N, Y =<N,
    black(Y,X), Y1 is Y+1,
    build_vrt(Y1,X,FDs).

% assoc_words(+ : F ilaDimList, − : F ilaList)
% P ost = {assoc_words(F Ds, F s) | useful({]F Ds[}) = {filadim(D_⋌ , Y_⋌ , X_⋌, L_⋌ ) | i = 1..n} ∧
% {]F s[} = {fila(D_⋌ , Y_⋌, X_⋌ , W s_⋌ ) | i = 1..n} ∧ {]W s_⋌[} = {w ∈ ws_0 | ‖w‖ = L_⋌ }}
% |assoc_words(F Ds, F s)| = ‖F Ds‖
assoc_words([],[]).
assoc_words([filadim(_,_,_,0) | FDs], Fs) :-
    assoc_words(FDs,Fs).
assoc_words([filadim(_,_,_,1) | FDs], Fs) :-
    assoc_words(FDs,Fs).
assoc_words([filadim(D,Y,X,L) | FDs], [fila(D,Y,X,Ws1) | Fs]) :-
    L > 1,
    assoc_words(FDs,Fs),
    words(Ws), words_of_ln(Ws,L,Ws1).

% words_of_ln(+ : AtomList, + : GaeN, − : AtomList)
% P ost = {words_of_ln(W s, L, W s1) | {]W s1[} = {w ∈ {]W s[} | ‖w‖ = L}}
% |words_of_ln(W s, L, W s1)| = ‖W s‖
words_of_ln([],_,[]).
words_of_ln([W | WS],L,WS1) :-
    atom_chars(W,W1),
    ln(W1,M), M =\= L,
    words_of_ln(WS,L,WS1).
words_of_ln([W | WS],L,[W | WS1]) :-
    atom_chars(W,W1),
    ln(W1,L),
    words_of_ln(WS,L,WS1).

% ordina_res(+ : ResList, − : AtomList)
% P ost = {ordina_res(Rs, W s) | {]Rs[} = {R_1, . . . , R_≪} ∧ R_⋌ <_® R_{⋌+1} ∧
% R_⋌ = res(D_⋌ , Y_⋌ , X_⋌, W_⋌ ) ∧ W s = [W_1, . . . , W_≪]}
ordina_res(Rs,Ws) :-
    qs_file(Rs,Rs1),
    only_words(Rs1,Ws).

% qs_file(+ : ResList, − : ResList)
% P ost = {qs_file(Rs, Rs1) | {]Rs[} = {R_1, . . ., R_≪} ∧ R_⋌ <_® R_{⋌+1} ∧ Rs1 = [R_1, . . . , R_≪]}
% |qs_file(Rs, Rs1)| = ‖Rs‖
qs_file([],[]).
qs_file([R | Rs], Rs1) :-
    part_file(R, Rs, Littles, Bigs),
    qs_file(Littles,LRs),

    qs_file(Bigs,BRs),
    app(LRs, [R | BRs], Rs1).

% part_file(+ : Res, + : ResList, − : ResList, − : ResList)
% P ost = {part_file(R, Rs, LRs, BRs) | {]Rs[} = {]LRs[} ∪ {]BRs[} ∧
% R1 ∈ LRs =⇒ R1 <_® R ∧ R1 ∈ BRs =⇒ R <_® R1}
% |part_file(R, Rs, LRs, BRs)| = ‖Rs‖
part_file(_,[],[],[]).
part_file(R,[R1 | Rs],[R1 | LRs],BRs) :-
    greater_fila(R,R1),
    part_file(R,Rs,LRs,BRs).
part_file(R,[R1 | Rs],LRs,[R1 | BRs]) :-
    \+ greater_fila(R,R1),
    part_file(R,Rs,LRs,BRs).

% greater_fila(+ : Res, + : Res)
% P ost = {greater_fila(R, R1) | R1 <_® R}
greater_fila(res(vrt,_,_,_),res(orz,_,_,_)).
greater_fila(res(orz,Y,_,_),res(orz,Y1,_,_)) :- Y > Y1.
greater_fila(res(orz,Y,X,_),res(orz,Y,X1,_)) :- X > X1.
greater_fila(res(vrt,_,X,_),res(vrt,_,X1,_)) :- X > X1.
greater_fila(res(vrt,Y,X,_),res(vrt,Y1,X,_)) :- Y > Y1.

% only_words(+ : ResList, − : AtomList)
% P ost = {only_words(Rs, W s) | Rs = [R_1, . . . , R_≪] ∧ R_⋌ = res(D_⋌ , Y_⋌, X_⋌ , W_⋌) ∧
% W s = [W_1, . . . , W_≪]}
% |only_words(Rs, W s)| = ‖Rs‖
only_words([],[]).
only_words([res(_,_,_,W)|Rs],[W|Ws]) :-
    only_words(Rs,Ws).

% build_cross(+ : F ilaList, + : Cross, − : ResList)
% P re = {build_cross(F s, C, Rs) | consistent({]C[}) ∧ compatible({]F s[}, {]C[})}
% P ost = {build_cross(F s, C, Rs) | consistent({]C[}) ∧ compatible({]F s[}, {]C[}) ∧
% {]Rs[} ∈ sol({]C[}, {]F s[})}
% |build_cross(F s, C, Rs)| = ‖F s‖
build_cross([],_,[]).
build_cross([F | Fs], C, [res(D,Y,X,W) | Rs]) :-
    min_fila([F | Fs], fila(D,Y,X,Ws), Fs1),
    member(W, Ws),
    assegna(D,Y,X,W,C,C1),
    aggiorna(Fs1,W,C1,Fs2),
    build_cross(Fs2,C1,Rs).

% assegna(+ : Dir, + : GaeN, + : GaeN, + : Atom, + : Cross, − : Cross)
% P ost = {assegna(D, Y, X, W, C, C1) | {]C1[} = assign({]C[}, W, D, Y, X)}
assegna(D,Y,X,W,C,C1) :-
    atom_chars(W,W1),
    assegna2(D,Y,X,W1,C,C1).

% assegna2(+ : Dir, + : GaeN, + : GaeN, + : CharList, + : Cross, − : Cross)
% P ost = {assegna2(D, Y, X, W, C, C1) | {]C1[} = assign({]C[}, W, D, Y, X)}

% |assegna2(D, Y, X, W, C, C1)| = ‖W ‖
assegna2(_,_,_,[],C,C).
assegna2(orz,Y,X,[Ch|W],C,[cell(Y,X,Ch) | C1]) :-
    X1 is X+1,
    assegna2(orz,Y,X1,W,C,C1).
assegna2(vrt,Y,X,[Ch|W],C,[cell(Y,X,Ch) | C1]) :-
    Y1 is Y+1,
    assegna2(vrt,Y1,X,W,C,C1).

% aggiorna(+ : F ilaList, + : Atom, + : Cross, − : F ilaList)
% P re = {aggiorna(F s, W, C, F s1) | consistent({]C[})}
% P ost = {aggiorna(F s, W, C, F s1) | consistent({]C[}) ∧ {]F s[} = {F_1, . . . , F_≪} ∧ {]F s1[} = {F^0_1, . . . F^0_≪} ∧
% F_⋌ = fila(D_⋌ , Y_⋌ , X_⋌, W s_⋌ ) ∧ F^0_⋌ = fila(D_⋌ , Y_⋌, X_⋌ , W s^0_⋌ ) ∧
% {]W s^0_⋌[} = filter(D_⋌ , Y_⋌, X_⋌ , {]W s_⋌[} − {W }, {]C[})}
% |aggiorna(F s, W, C, F s1)| = ‖F s‖
aggiorna([],_,_,[]).
aggiorna([fila(D,Y,X,Ws)|Fs1],W,C,[fila(D,Y,X,Ws2)|Fs2]) :-
    minus_el(Ws,W,Ws1),
    filtra_cross(D,Y,X,Ws1,C,Ws2),
    aggiorna(Fs1,W,C,Fs2).

% minus_el(+ : AtomList, + : Atom, − : AtomList)
% P ost = {minus_el(W s, W, W s1) | {]W s1[} = {]W s[} − {W }}
% |minus_el(W s, W, W s1)| = ‖W s‖
minus_el([],_,[]).
minus_el([W|Ws],W,Ws).
minus_el([W|Ws],W1,[W|Ws1]) :-
    W \== W1,
    minus_el(Ws,W1,Ws1).

% filtra_cross(+ : Dir, + : GaeN, + : GaeN, + : AtomList, + : Cross, − : AtomList)
% P re = {filtra_cross(D, Y, X, W s, C, W s1) | consistent({]C[})}
% P ost = {filtra_cross(D, Y, X, W s, C, W s1) | consistent({]C[}) ∧ {]W s1[} = filter(D, Y, X, {]W s[}, {]C[})}
% |filtra_cross(D, Y, S, W s, C, W s1)| = ‖W s‖
filtra_cross(_,_,_,[],_,[]).
filtra_cross(D,Y,X,[W|Ws],C,[W|Ws1]) :-
    atom_chars(W,W1),
    filtra_cross_aux(D,Y,X,W1,C),
    filtra_cross(D,Y,X,Ws,C,Ws1).
filtra_cross(D,Y,X,[W|Ws],C,Ws1) :-
    atom_chars(W,W1),
    \+ filtra_cross_aux(D,Y,X,W1,C),
    filtra_cross(D,Y,X,Ws,C,Ws1).

% filtra_cross_aux(+ : Dir, + : GaeN, + : GaeN, + : CharList, + : Cross)
% P re = {filtra_cross_aux(D, Y, X, W, X) | consistent({]C[})}
% P ost = {filtra_cross_aux(D, Y, X, W, C) | consistent({]C[}) ∧ filter(D, Y, X, {W }, {]C[}) = {W }}
% |filtra_cross_aux(D, Y, S, W, C)| = ‖W ‖
filtra_cross_aux(_,_,_,[],_).
filtra_cross_aux(orz,Y,X,[Ch|W],C) :-
    possibly_in(Y,X,Ch,C),
    X1 is X+1,

    filtra_cross_aux(orz,Y,X1,W,C).
filtra_cross_aux(vrt,Y,X,[Ch|W],C) :-
    possibly_in(Y,X,Ch,C),
    Y1 is Y+1,
    filtra_cross_aux(vrt,Y1,X,W,C).

% possibly_in(+ : GaeN, + : GaeN, + : Char, + : Cross)
% P re = {possibly_in(Y, X, Ch, C) | consistent({]C[})}
% P ost = {possibly_in(Y, X, Ch, C) | consistent({]C[}) ∧ cell(Y, X, Ch1) ∈ {]C[} =⇒ Ch1 = Ch}
% |possibly_in(Y, X, Ch, C)| = ‖C‖
possibly_in(Y,X,Ch,[cell(Y,X,Ch)|_]).
possibly_in(_,_,_,[]).
possibly_in(Y,X,Ch,[cell(Y1,_,_)|C]) :-
    Y =\= Y1,
    possibly_in(Y,X,Ch,C).
possibly_in(Y,X,Ch,[cell(Y,X1,_)|C]) :-
    X =\= X1,
    possibly_in(Y,X,Ch,C).

% min_fila(+ : F ilaList, − : F ila, − : F ilaList)
% P ost = {min_fila(F s, F, F s1) | {]F s[} = {]F s1[} ∪ {F }}
% |min_fila(F s, F, F s1)| = ‖F s‖
min_fila([F],F,[]).
min_fila([F|Fs],M,[F|Fs1]) :-
    min_fila(Fs,M,Fs1),
    shorter_fila(M,F).
min_fila([F|Fs],F,[M|Fs1]) :-
    min_fila(Fs,M,Fs1),
    \+ shorter_fila(M,F).

% shorter_fila(+ : F ila, + : F ila)
% P ost = {shorter_fila(fila(D, Y, X, W s), fila(D^0, Y^0, X^0, W s1)) | ‖W s‖ ≤ ‖W s1‖}
shorter_fila(fila(_,_,_,Ws),fila(_,_,_,Ws1)) :-
    ln(Ws,L), ln(Ws1,L1), L =< L1.

% ln(+ : List, − : GaeN )
% P ost = {ln(Ls, L) | L = ‖Ls‖}

% member(− : Atom, + : AtomList)
% P ost = {member(A, As) | A ∈ {]As[}}

% app(− : T ermList, − : T ermList, + : T ermList)
% P ost = {app(Ls1, Ls2, Ls) | Ls = Ls1 ⋆ Ls2}

% atom_char(+ : Atom, − : CharList)
% P ost = {atom_char(W, W 1) | W = W 1}

% is(− : GaeN, + : GaeN )
% P ost = {is(N, M ) | N = M }

% =<(+ : GaeN, + : GaeN )
% P ost = {=<(N, M ) | N ≤ M }

% <(+ : GaeN, + : GaeN )

% P ost = {<(N, M ) | N < M }

% =:=(+ : GaeN, + : GaeN )
% P ost = {=:=(N, M ) | N = M }

% =\=(+ : GaeN, + : GaeN )
% P ost = {=\=(N, M ) | N ≠ M }

% \==(+ : T erm, + : T erm)
% P ost = {\==(T, T )}

3 Weak correctness

It is easy to see that:

Proposition 1 The program is well-moded, well-typed, output-linear and non-floundering.

This is sufficient to prove that:

Corollary 2

• The occur check can be safely omitted.

• The program does not end in error.

Partial correctness and termination of the program can be derived from the following proof outlines (only the interesting predicates are presented). {true}

crossword(Ws) :-

{true}

build_file(Fs),

{{R_1, . . . , R_≪} ∈ sol(∅, {]F s[}), R_⋌ <_® R_{⋌+1}, R_⋌ = res(D_⋌ , Y_⋌, X_⋌ , W_⋌) =⇒ [W_1, . . . , W_≪] is a solution of P_0} {true}

words(Ws1),

{{]W s1[} = ws_0} {true}

ln(Ws1,L),

{L = ‖W s1‖} {true}

ln(Fs,L),

{L = ‖F s‖} {consistent(∅) ∧ compatible({]F s[}, ∅)}

build_cross(Fs,[],Rs),

{consistent(∅) ∧ compatible({]F s[}, ∅) ∧ {]Rs[} ∈ sol(∅, {]F s[})} {true}

ordina_res(Rs,Ws).

{{]Rs[} = {R_1, . . . , R_≪} ∧ R_⋌ <_® R_{⋌+1} ∧ R_⋌ = res(D_⋌ , Y_⋌, X_⋌ , W_⋌) ∧ W s = [W_1, . . . , W_≪ ]} {W s is an exact solution of P_0}

Notice that if [W_1, . . . , W_≪] is a solution of P_0 and n = |ws_0| then [W_1, . . . , W_≪] is an exact solution of P_0. {true}

build_file(Fs) :-

{1 ∈ [1..n_0 + 1]}

build_orz(1,1,FDs1),

{F Ds1 = orz(C_0, 1, 1)} {1 ∈ [1..n_0 + 1]}

build_vrt(1,1,FDs2),

{F Ds2 = vrt(C_0, 1, 1)} {true}

app(FDs1,FDs2,FDs),

{F Ds = F Ds1 ⋆ F Ds2} {true}

assoc_words(FDs,Fs).

{{]F Ds[} = {filadim(D_⋌ , Y_⋌ , X_⋌, L_⋌ ) | i = 1..n} ∧ {]F s[} = fila(D_⋌ , Y_⋌ , X_⋌, W s_⋌ ) | i = 1..n} ∧ {]W s_⋌ [} = {w ∈ ws_0 | ‖w‖ = L_⋌ }} {R_1, . . . , R_≪} ∈ sol(∅, {]F s[}), R_⋌ <_® R_{⋌+1}, R_⋌ = res(D_⋌ , Y_⋌ , X_⋌, W_⋌ ) =⇒ [W_1, . . . , W_≪] is a solution of P_0}

The previous proof outline is based on the following lemma, whose proof is difficult to formalize, although it is conceptually simple.

Lemma 3 Suppose that: 1. useful({]orz(C_0 , 1, 1)[} ∪ {]vrt(C_0, 1, 1)[}) = {filadim(D_⋌ , Y_⋌, X_⋌ , L_⋌ ) | i = 1..n}, 2. {]W s_⋌ [} = {w ∈ ws_0 | ‖w‖ = L_⋌ }, and

3. {R_1, . . . , R_≪} ∈ sol(∅, {]F s[}) with R_⋌ <_® R_{⋌+1} and R_⋌ = res(D_⋌ , Y_⋌ , X_⋌, W_⋌ ).

Then [W_1, . . . , W_≪] is a solution of P_0. {M, X ∈ [1..n_0 + 1]}

build_orz(M,X,[filadim(orz,M,X,0)]) :- {(n_0 + 1 − X) + (n_0 + 1 − M )(n_0 + 1)}

{true}

dimension(N),

{N = n_0} {true}

M =:= N+1.

{M = n_0 + 1} {n_0 + 1, X ∈ [1..n_0 + 1] ∧ [filadim(orz, n_0 + 1, X, 0)] = orz(C_0, n_0 + 1, X)} {Y, M ∈ [1..n_0 + 1]}

build_orz(Y,M,[filadim(orz,Y,M,0) | FDs]) :- {(n_0 + 1 − M ) + (n_0 + 1 − Y )(n_0 + 1)}

{true}

dimension(N),

{N = n_0} {true}

M =:= N+1,

{M = n_0 + 1} {true}

Y =< N,

{Y ≤ n_0} {true}

Y1 is Y+1,

{Y 1 = Y + 1} {Y + 1, 1 ∈ [1..n_0 + 1]}

build_orz(Y1,1,FDs). {n_0 + (n_0 − Y )(n_0 + 1)}

{Y + 1, 1 ∈ [1..n_0 + 1] ∧ F Ds = orz(C_0, Y + 1, 1)} {Y, n_0 + 1 ∈ [1..n_0 + 1] ∧ [filadim(orz, Y, n_0 + 1, 0)|F Ds] = orz(C_0, Y, n_0 + 1)} {Y, X ∈ [1..n_0 + 1]}

build_orz(Y,X,[filadim(orz,Y,X,L1) | FDs]) :- {(n_0 + 1 − X) + (n_0 + 1 − Y )(n_0 + 1)}

{true}

dimension(N),

{N = n_0} {true}

X =< N,

{X ≤ n_0} {true}

Y =< N,

{Y ≤ n_0} {true}

\+black(Y,X),

{(Y, X) ∉ black_0} {true}

X1 is X+1,

{X1 = X + 1} {Y, X + 1 ∈ [1..n_0 + 1]}

build_orz(Y,X1,[filadim(D,W,Z,L) | FDs]), {(n_0 − X) + (n_0 + 1 − Y )(n_0 + 1)}

{Y, X + 1 ∈ [1..n_0 + 1] ∧ [filadim(orz, W, Z, L)|F Ds] = orz(C_0, Y, X + 1)} {true}

L1 is L+1.

{L1 = L + 1} {Y, X ∈ [1..n_0 + 1] ∧ [filadim(orz, Y, X, L + 1)|F Ds] = orz(C_0, Y, X)} {Y, X ∈ [1..n_0 + 1]}

build_orz(Y,X,[filadim(orz,Y,X,0) | FDs]) :- {(n_0 + 1 − X) + (n_0 + 1 − Y )(n_0 + 1)}

{true}

dimension(N),

{N = n_0} {true}

X =< N,

{X ≤ n_0} {true}

Y =< N,

{Y ≤ n_0} {true}

black(Y,X),

{(Y, X) ∈ black_0} {true}

X1 is X+1,

{X1 = X + 1} {Y, X + 1 ∈ [1..n_0 + 1]}

build_orz(Y,X1,FDs). {(n_0 − X) + (n_0 + 1 − Y )(n_0 + 1)}

{Y, X + 1 ∈ [1..n_0 + 1] ∧ F Ds = orz(C_0, Y, X + 1)} {Y, X ∈ [1..n_0 + 1] ∧ [filadim(orz, Y, X, 0)|F Ds] = orz(C_0, Y, X)}

{true}

assoc_words([],[]). {0}

{{][][} = ∅}

{true}

assoc_words([filadim(D,Y,X,0) | FDs],Fs) :- {‖[F Ds]‖ + 1}

{true}

assoc_words(FDs,Fs). {‖F Ds‖}

{useful({]F Ds[}) = {filadim(D_⋌ , Y_⋌, X_⋌ , L_⋌) | i = 1..n} ∧ {]F s[} = {fila(D_⋌ , Y_⋌ , X_⋌, W s_⋌ ) | i = 1..n} ∧ {]W s_⋌ [} = {w ∈ ws_0 | ‖w‖ = L_⋌ }} {useful({][filadim(D, Y, X, 0)|F Ds][}) = {filadim(D_⋌ , Y_⋌, X_⋌ , L_⋌) | i = 1..n} ∧ {]F s[} = {fila(D_⋌ , Y_⋌, X_⋌ , W s_⋌ ) | i = 1..n} ∧ {]W s_⋌[} = {w ∈ ws_0 | ‖w‖ = L_⋌ }} {true}

assoc_words([filadim(D,Y,X,1) | FDs],Fs) :- {‖F Ds‖ + 1}

{true}

assoc_words(FDs,Fs). {‖F Ds‖}

{useful({]F Ds[}) = {filadim(D_⋌ , Y_⋌, X_⋌ , L_⋌) | i = 1..n} ∧ {]F s[} = {fila(D_⋌ , Y_⋌ , X_⋌, W s_⋌ ) | i = 1..n} ∧ {]W s_⋌ [} = {w ∈ ws_0 | ‖w‖ = L_⋌ }} {useful({][filadim(D, Y, X, 1)|F Ds][}) = {filadim(D_⋌ , Y_⋌, X_⋌ , L_⋌) | i = 1..n} ∧ {]F s[} = {fila(D_⋌ , Y_⋌, X_⋌ , W s_⋌ ) | i = 1..n} ∧ {]W s_⋌[} = {w ∈ ws_0 | ‖w‖ = L_⋌ }} {true}

assoc_words([filadim(D,Y,X,L) | FDs],[fila(D,Y,X,Ws1) | Fs]) :- {‖F Ds‖ + 1}

{true}

1 < L,

{1 < L} {true}

assoc_words(FDs,Fs), {‖F Ds‖}

{useful({]F Ds[}) = {filadim(D_⋌ , Y_⋌, X_⋌ , L_⋌) | i = 1..n} ∧ {]F s[} = {fila(D_⋌ , Y_⋌ , X_⋌, W s_⋌ ) | i = 1..n} ∧ {]W s_⋌ [} = {w ∈ ws_0 | ‖w‖ = L_⋌ }} {true}

words(Ws),

{{]W s[} = ws_0} {true}

words_of_ln(Ws,L,Ws1).

{{]W s1[} = {w ∈ {]W s[} | ‖w‖ = L}} {useful({][filadim(D, Y, X, L)|F Ds][}) = {filadim(D, Y, X, L)} ∪ {filadim(D_⋌ , Y_⋌ , X_⋌, L_⋌ ) | i = 1..n} ∧ {][fila(D, Y, X, W s1)|F s][} = {fila(D, Y, X, W s1)} ∪ {fila(D_⋌ , Y_⋌, X_⋌ , W s_⋌ ) | i = 1..n} ∧ {]W s_⋌[} = {w ∈ ws_0 | ‖w‖ = L_⋌ } ∧ {]W s1[} = {w ∈ ws_0 | ‖w‖ = L}}

{true}

ordina_res(Rs,Ws) :-

{true}

qs_file(Rs,Rs1),

{{]Rs[} = {R_1, . . . , R_≪} ∧ R_⋌ <_® R_{⋌+1} ∧ Rs1 = [R_1, . . . , R_≪]} {true}

only_words(Rs1,Ws).

{Rs1 = [R_1, . . . , R_≪] ∧ R_⋌ = res(D_⋌ , Y_⋌, X_⋌ , W_⋌) ∧ W s = [W_1, . . ., W_≪ ]} {{]Rs[} = {R_1, . . . , R_≪} ∧ R_⋌ <_® R_{⋌+1} ∧ R_⋌ = res(D_⋌ , Y_⋌ , X_⋌, W_⋌ ) ∧ W s = [W_1, . . . , W_≪]} {true}

only_words([],[]). {0}

{[] = []}

{true}

only_words([res(D,Y,X,W) | Rs],[W | Ws]) :- {‖Rs‖ + 1}

{true}

only_words(Rs,Ws). {‖Rs‖}

{Rs = [R_1, . . . , R_≪] ∧ R_⋌ = res(D_⋌ , Y_⋌, X_⋌ , W_⋌) ∧ W s = [W_1, . . . , W_≪]} {[res(D, Y, X, W )|Rs] = [res(D, Y, X, W ), R_1, . . . , R_≪] ∧ R_⋌ = res(D_⋌ , Y_⋌ , X_⋌, W_⋌ ) ∧ [W |W s] = [W, W_1, . . . , W_≪]}

{consistent({]C[}) ∧ compatible(∅, {]C[})}

build_cross([],C,[]). {0}

{consistent({]C[}) ∧ compatible(∅, {]C[}) ∧ ∅ ∈ sol({]C[}, ∅)} {consistent({]C[}) ∧ compatible({F } ∪ {]F s[}, {]C[}}

build_cross([F | Fs], C, [res(D,Y,X,W) | Rs]) :- {‖F s‖ + 1}

{true}

min_fila([F | Fs], fila(D,Y,X,Ws), Fs1),

{{F } ∪ {]F s[} = {fila(D, Y, X, W s)} ∪ {]F s1[}} {true}

member(W, Ws),

{W ∈ {]W s[}} {true}

assegna(D,Y,X,W,C,C1),

{{]C1[} = assign({]C[}, W, D, Y, X)} {consistent({]C1[})}

aggiorna(Fs1,W,C1,Fs2),

{consistent({]C1[}) ∧ {]F s1[} = {F_1, . . . , F_≪} ∧ {]F s2[} = {F^0_1, . . . F^0_≪} ∧ F_⋌ = fila(D_⋌ , Y_⋌ , X_⋌, W s_⋌ ) ∧ F^0_⋌ = fila(D_⋌ , Y_⋌ , X_⋌, W s^0_⋌ ) ∧ {]W s^0_⋌ [} = filter(D_⋌ , Y_⋌, X_⋌ , {]W s_⋌[} − {W }, C1)} {consistent({]C1[}) ∧ compatible({]F s2[}, {]C1[})}

build_cross(Fs2,C1,Rs). {‖F s2‖}

{consistent({]C1[}) ∧ compatible({]F s2[}, {]C1[}) ∧ {]Rs[} ∈ sol({]C1[}, {]F s2[})} {consistent({]C[}) ∧ compatible({F } ∪ {]F s[}, {]C[}) ∧ {res(D, Y, X, W )} ∪ {]Rs[} ∈ sol({]C[}, {F } ∪ {]F s[})}

The following lemmas have been used in this proof outline. Lemma 4 Let SC be a set of Cell, D ∈ Dir, Y, X ∈ GaeN and W s = [W_1, . . . , W_≪] ∈ AtomList. If consistent(SC) and compatible({fila(D, Y, X, W s)}, SC) then consistent(assign(SC, W_⋌ , D, Y, X)) for i = 1..n. Lemma 5 Let W s ∈ AtomList, D ∈ Dir, Y, X ∈ GaeN , SW be a set of Atom and SC be a set of Cell. Then: {]W s[} = filter(D, Y, X, SW, SC) =⇒ compatible({fila(D, Y, X, W s)}, SC). Notice that compatible(SF, SC) iff compatible({F }, SC) for each F ∈ SF . {true}

assegna(D,Y,X,W,C,C1) :-

{true}

atom_chars(W,W1),

{W = W 1} {true}

assegna2(D,Y,X,W1,C,C1).

{{]C1[} = assign({]C[}, W 1, D, Y, X)} {{]C1[} = assign({]C[}, W, D, Y, X)}

{true}

assegna2(D,Y,X,[],C,C). {0}

{{]C[} = assign({]C[}, [], D, Y, X)} {true}

assegna2(orz,Y,X,[Ch|W],C,[cell(Y,X,Ch) | C1]) :- {‖W ‖ + 1}

{true}

X1 is X+1,

{X1 = X + 1} {true}

assegna2(orz,Y,X1,W,C,C1). {‖W ‖}

{{]C1[} = assign({]C[}, W, orz, Y, X + 1)} {{cell(Y, X, Ch)} ∪ {]C1[} = assign({]C[}, [Ch|W ], orz, Y, X)}

{consistent({]C[})}

aggiorna([],W,C,[]). {0}

{consisten({]C[}) ∧ {][][} = ∅} {consistent({]C[})}

aggiorna([fila(D,Y,X,Ws)|Fs1],W,C,[fila(D,Y,X,Ws2)|Fs2]) :- {‖F s1‖ + 1}

{true}

minus_el(Ws,W,Ws1),

{{]W s1[} = {]W s[} − {W }} {consistent({]C[})}

filtra_cross(D,Y,X,Ws1,C,Ws2),

{consistent({]C[}) ∧ {]W s2[} = filter(D, Y, X, {]W s1[}, {]C[})} {consistent({]C[})}

aggiorna(Fs1,W,C,Fs2). {‖F s1‖}

{consistent({]C[}) ∧ {]F s1[} = {F_1, . . . , F_≪} ∧ {]F s2[} = {F^0_1, . . . F^0_≪ } ∧ F_⋌ = fila(D_⋌ , Y_⋌ , X_⋌, W s_⋌ ) ∧ F^0_⋌ = fila(D_⋌ , Y_⋌ , X_⋌, W s^0_⋌ ) ∧ {]W s^0_⋌ [} = filter(D_⋌ , Y_⋌, X_⋌ , {]W s_⋌[} − {W }, {]C[})} {consistent({]C[}) ∧ {][fila(D, Y, X, W s)|F s1][} = {fila(D, Y, X, W s), F_1, . . . , F_≪} ∧ {][fila(D, Y, X, W s2)|F s2][} = {fila(D, Y, X, W s2), F^0_1, . . . F^0_≪} ∧ F_⋌ = fila(D_⋌ , Y_⋌ , X_⋌, W s_⋌ ) ∧ F^0_⋌ = fila(D_⋌ , Y_⋌, X_⋌ , W s^0_⋌ ) ∧ {]W s^0_⋌[} = filter(D_⋌ , Y_⋌, X_⋌ , {]W s_⋌[} − {W }, {]C[}) ∧ {]W s2[} = filter(D, Y, X, {]W s[} − {W }, {]C[})}

{consistent({]C[})}

filtra_cross(D,Y,X,[],C,[]). {0}

{consistent(C) ∧ filter(D, Y, X, ∅, {]C[}) = ∅} {consistent({]C[})}

filtra_cross(D,Y,X,[W|Ws],C,[W|Ws1]) :- {‖W s‖ + 1}

{true}

atom_chars(W,W1),

{W = W 1} {consistent({]C[})}

filtra_cross_aux(D,Y,X,W1,C),

{consistent({]C[}) ∧ filter(D, Y, X, {W 1}, {]C[}) = {W 1}}

{consistent({]C[})}

filtra_cross(D,Y,X,Ws,C,Ws1). {‖W s‖}

{consistent({]C[}) ∧ {]W s1[} = filter(D, Y, X, {]W s[}, {]C[})} {consistent({]C[}) ∧ {][W |W s1][} = filter(D, Y, X, {][W |W s][}, {]C[})} {consistent({]C[})}

filtra_cross(D,Y,X,[W|Ws],C,Ws1) :- {‖W s‖ + 1}

{true}

atom_chars(W,W1),

{W 1 = W } {consistent({]C[})}

\+ filtra_cross_aux(D,Y,X,W1,C),

{consistent({]C[}) ∧ filter(D, Y, X, {W 1}, {]C[}) = ∅} {consistent({]C[})}

filtra_cross(D,Y,X,Ws,C,Ws1). {‖W s‖}

{consistent({]C[}) ∧ {]W s1[} = filter(D, Y, X, {]W s[}, {]C[})} {consistent({]C[}) ∧ {]W s1[} = filter(D, Y, X, {][W |W s][}, {]C[})}

{consistent({]C[})}

filtra_cross_aux(D,Y,X,[],C). {0}

{consistent({]C[}) ∧ filter(D, Y, X, {[]}, {]C[}) = {[]}} {consistent({]C[})}

filtra_cross_aux(orz,Y,X,[Ch|W],C) :- {‖W ‖ + 1}

{consistent({]C[})}

possibly_in(Y,X,Ch,C),

{consistent({]C[}) ∧ cell(Y, X, Ch1) ∈ {]C[} =⇒ Ch1 = Ch} {true}

X1 is X+1,

{X1 = X + 1} {consistent({]C[})}

filtra_cross_aux(orz,Y,X1,W,C). {‖W ‖}

{consistent({]C[}) ∧ filter(orz, Y, X + 1, {W }, {]C[}) = {W }} {consistent({]C[}) ∧ filter(orz, Y, X, {[Ch|W ]}, {]C[}) = {[Ch|W ]}}

Notice that both filter(orz, Y, X +1, {W }, SC) = {W } and cell(Y, X, Ch1) ∈ SC =⇒ Ch1 = Ch hold if and only if filter(orz, Y, X, {[Ch|W ]}, SC) = {[Ch|W ]}. {consistent(cell(Y, X, Ch) ∪ {]C[})}

possibly_in(Y,X,Ch,[cell(Y,X,Ch)|C]). {‖C‖ + 1}

{consistent(cell(Y, X, Ch) ∪ {]C[}) ∧

cell(Y, X, Ch1) ∈ {cell(Y, X, Ch)} ∪ {]C[} =⇒ Ch1 = Ch}

{consistent(∅)}

possibly_in(Y,X,Ch,[]). {0}

{consistent(∅)} {consistent({cell(Y 1, X1, Ch1)} ∪ {]C[})}

possibly_in(Y,X,Ch,[cell(Y1,X1,Ch1)|C]) :- {‖C‖ + 1}

{true}

Y =\= Y1,

{Y ≠ Y 1} {consistent({]C[})} possibly_in(Y,X,Ch,C). {‖C‖}

{consistent({]C[}) ∧ cell(Y, X, Ch^0) ∈ {]C[} =⇒ Ch^0 = Ch} {consistent({cell(Y 1, X1, Ch1)} ∪ {]C[}) ∧

cell(Y, X, Ch^0) ∈ {cell(Y 1, X1, Ch1)} ∪ {]C[} =⇒ Ch^0 = Ch}

Notice that, if consistent(SC) and cell(Y, X, Ch) ∈ SC then cell(Y, X, Ch^0) ∈ SC =⇒ Ch^0 = Ch. {true}

min_fila([F],F,[]). {1}

{{F } = {F }} {true}

min_fila([F|Fs],M,[F|Fs1]) :- {‖F s‖ + 1}

{true}

min_fila(Fs,M,Fs1), {‖F s‖}

{{]F s[} = {M } ∪ {]F s1[}} {true}

shorter_fila(M,F).

{M = fila(D, Y, X, W s) ∧ F = fila(D^0, Y^0, X^0, W s^0) ∧ ‖W s‖ ≤ ‖W s^0‖} {{]F s[} ∪ {F } = {M, F } ∪ {]F s1[}} {true}

min_fila([F|Fs],F,[M|Fs1]) :- {‖F s‖ + 1}

{true}

min_fila(Fs,M,Fs1), {‖F s‖}

{{]F s[} = {M } ∪ {]F s1[}} {true}

\+ shorter_fila(M,F).

{M = fila(D, Y, X, W s) ∧ F = fila(D^0, Y^0, X^0, W s^0) ∧ ‖W s‖ > ‖W s^0‖} {{]F s[} ∪ {F } = {M, F } ∪ {]F s1[}}

Theorem 6

• The program is weakly partial correct.

• The program terminates.

4 About (strong) correctness

The program is also (strongly) totally correct. In fact, each solution of the puzzle is found by the program. However, to prove this we should exhibit a model supported by the program. There are some difficulties, which arise since in the program we extensively use lists to represent (multi)sets. A set can be represented by several lists which differ only in the order of the elements and during the esecution of the program just one of these lists is generated to represent a particular set. To have a supported model, we should exactly specify the representative list, whereas the intended semantics specifies only the corresponding set.

For instance, consider the predicate minmax, that, given a set of integer, returns its minimum element, its maximum element and the rest of the set:

minmax([M,N],M,N,[]) :-
    M =< N.
minmax([N,M],M,N,[]) :-

    N > M.
minmax([L|Ns],M,N,[L|Ns1]) :-
    minmax(Ns,M,N,Ns1),
    M =< L, L =< N.
minmax([L|Ns],L,N,[M|Ns1]) :-
    minmax(Ns,M,N,Ns1),
    L < M.
minmax([L|Ns],M,L,[N|Ns1]) :-
    minmax(Ns,M,N,Ns1),
    L > N.

The intended semantics is given by the pre- and post-conditions P re = {minmax(N s, M, N, N s1) | N s ∈ GaeN List ∧ ‖N s‖ ≥ 2} P ost = {minmax(N s, M, N, N s1) | {]N s[} = {]N s1[} ∪ {M, N } ∧ M = min{]N s[} ∧ N = max{]N s[}} However, to obtain a supported model, we should specify exactly how list N s1 is obtained from N s. This is not easy: for instance

minmax([1, 8, 7, 2, 6, 5, 4, 3], M, N, N s1)

returns:

M = 1, N = 8, N s1 = [2, 7, 6, 3, 5, 4]

In our opinion, the “correct” post-condition does not respect the intended semantics, since it specifies uninteresting implemention details.

This particular procedure could be easily changed. However, lists are extensively used in our program to represent sets, and it would be very boring to define standard representatives for all those sets.

As a general issue, it would be important to have techniques to deal with abstract data types: in this case it would possible to consider lists which represent sets just as “sets” (and to avoid the heavy usage of {]List[}).

The necessity to give a supported model is a too strong condition also from another point of view: in proving correctness, in fact, it is useful to abstract away from implementation details, that can improve efficiency, but that are not meaningful for correctnes.

For instance, in the definition of the post-condition of min_fila(F s, F, F s1) we do not care about the fact that F is the smallest F ila of F s. For build_cross to be correct, in fact, it is only important that F is an element of F s. The choice of the minimal element is important only to have a more efficient program.

5 Parallelizability via delay declarations

The program is intrinsecally sequential. Some clauses could however be parallelized, in particular those which map a function on a set (assoc_words, aggiorna) or which filter a set (filtra_cross), or the routines like qs which have already be analized in the literature. The techniques we know to deal with parallelizability via delay declaretions cannot be applied to the program.